
      ! ***********************************************************************
      ! *                                                                     *
      ! *  ** Meilhaus SDK example for ME-4000 / ME-6000 ***                  *
      ! *                                                                     *
      ! *  Purpose: Single analog input for ME-4000 series                    *
      ! *                                                                     *
      ! *  (c) Tech Soft GmbH, 2006                                           *
      ! *                                                                     *
      ! *  Created:  10-Jan-2006                                              *
      ! *  Updated:  01-Jun-2006                                              *
      ! *                                                                     *
      ! ***********************************************************************
      !
      ! This program uses the lib file "me_sdk_for_htb.lib" which contains
      ! several definitions and functions in order to make it easier to
      ! talk to the Meilhaus ME-4000 / ME-6000 hardware.
      !
      DLL UNLOAD ALL
      CLEAR SCREEN
      !
      CONTROL CRT,100;1
      !
      PRINT "Test program for Meilhaus Interfaces"
      PRINT "------------------------------------"
      PRINT
      !
      ! variables
      LONG Dll_vers,Drv_vers,Lboardnum,Serialnum
      INTEGER Iboardnum,Ichannum,S_digital_val
      INTEGER Ret1,Ret2,Result
      LONG Ai_range,Ai_input,Ai_sdmode,Ai_trigger,Val_not_used
      REAL D_volt
      !
      ! Program messages on/off. 0=off, 1=show all messages, 2=show errors only
      !                          3=show debug messages
      INTEGER Verbose=2
      !
      G$=CHR$(19)                      ! character code for inverse printing
      X$=CHR$(16)                      ! character code for normal printing
      Hl$=CHR$(30)                     ! character code for "highlighted" printing
      Hx$=CHR$(24)                     ! character code for "normal" printing (color)
      Ul$=CHR$(22)                     ! character code for underlined printing
      Crlf$=CHR$(10)&CHR$(13)
      !
      ! Load the function library from Tech Soft for usage of ME cards
      DIM Libname$[255]
      ON ERROR GOTO Load_sub
      DELSUB Me_lib_start TO END
 Load_sub: OFF ERROR
      ON ERROR GOTO No_subs
      Libname$="..\meilhaus_for_htb.lib"   ! name of HTBasic library file for ME SDK
      IF NOT INMEM("Me_lib_start") THEN LOADSUB ALL FROM Libname$
      GOTO Weiter
      !
 No_subs: !
      BEEP
      PRINTER IS CRT
      ALPHA PEN 2
      PRINT "** ERROR: Can't find SUB programs. Please check your path."
      ALPHA PEN 1
      PRINT " * Missing file: "&SYSTEM$("MSI")&"\"&Libname$
      PRINT
      PRINT " Program stopped."
      STOP
      !
 Weiter: OFF ERROR
      ! initialize the library
      Res1=FNMe_init("me4000",Verbose)   ! initialize library for series ME-4000
      IF NOT Res1 THEN
          IF Verbose>0 THEN

              BEEP
              ALPHA PEN 2
              PRINT "** ERROR: ";
              ALPHA PEN 1
              PRINT "FNMe_init() failed, program stopped."
          END IF
          GOSUB Finish
          STOP
      END IF
      !
      ! get DLL version
      Ret1=FNM4getdllversion(Dll_vers)
      CALL Me_show_errstat(Ret1,"FNM4getdllversion()",Verbose)
      !
      Ret1=FNM4getdrvversion(Drv_vers)
      CALL Me_show_errstat(Ret1,"FNM4getdrvversion()",Verbose)
      !
      ! get serial number of board 0
      Boardnum=0
      Ret1=FNM4getserialnum((Boardnum),Serialnum)
      CALL Me_show_errstat(Ret1,"FNM4getserialnum()",Verbose)
      !
      PRINT " * DLL: 0x"&IVAL$(Dll_vers,16)&"   Driver: 0x"&IVAL$(Drv_vers,16);
      PRINT "   Serial number: 0x"&IVAL$(Serialnum,16)
      PRINT
      !
      ! now start one measure on board 0, channel 0
      Iboardnum=0
      Ichannum=0
      !
      PRINT G$&" Read one analog value (Board "&VAL$(Iboardnum);
      PRINT ", Chn. "&VAL$(Ichannum)&") "&X$
      !
      ! define constants
      Ai_range=FNMe_get_constant("ME4000_AI_RANGE_UNIPOLAR_10",Result,Verbose)
      Ai_sdmode=FNMe_get_constant("ME4000_AI_INPUT_SINGLE_ENDED",Result,Verbose)
      Ai_trigger=FNMe_get_constant("ME4000_AI_TRIGGER_SOFTWARE",Result,Verbose)
      Val_not_used=FNMe_get_constant("ME4000_VALUE_NOT_USED",Result,Verbose)
      !
      ! call the function
      Ret1=FNM4aisingle((Iboardnum),(Ichannum),(Ai_range),(Ai_sdmode),(Ai_trigger),(Val_
not_used),(Val_not_used),S_digital_val)
      CALL Me_show_errstat(Ret1,"FNM4aisingle()",Verbose)
      !
      IF Ret1=0 THEN         ! convert digital value into physical value
          Ret2=FNM4aidigittovolt((S_digital_val),(Ai_range),D_volt)
          CALL Me_show_errstat(Ret2,"FNM4aidigittovolt()",Verbose)
      END IF
      !
      PRINT " * Digital value: ";IVAL$(S_digital_val,16);"  physical value: ";D_volt;" V"
      PRINT
      !
      GOSUB Finish
      !
      STOP
      !
      !================================================================================
 Finish: ! always jump here if program ends
      DLL UNLOAD ALL
      !
      ! cleaning up SUB programs
      DELSUB Me_lib_start TO END
      ALPHA PEN 4
      DISP "Program finished."
      ALPHA PEN 1
      RETURN
      !
      !================================================================================
      END


